

    

        Finitomata

        v0.27.0


          [image: Logo]



    


  

    Table of contents

    
      



      	Finitomata


      	Finite Automata


      	Finitomata Compiler





        	
          
            	Modules
              


    	FSM
      


      	Finitomata


      	Finitomata.Accessible


      	Infinitomata



      

    




    	Test
      


      	Finitomata.ExUnit



      

    




    	Goods
      


      	Finitomata.Cache


      	Finitomata.Pool


      	Finitomata.Throttler



      

    




    	Internals
      


      	Finitomata.ClusterInfo


      	Finitomata.Listener


      	Finitomata.Parser


      	Finitomata.Pool.Actor


      	Finitomata.State


      	Finitomata.Supervisor


      	Finitomata.Transition


      	Finitomata.Transition.Path



      

    




    	Persistence
      


      	Finitomata.Persistency


      	Finitomata.Persistency.Persistable


      	Finitomata.Persistency.Protocol



      

    




    	Exceptions
      


      	Finitomata.TestSyntaxError


      	Finitomata.TestTransitionError



      

    



              

            

          

        



  	Mix Tasks
    

    	mix finitomata.generate


    	mix finitomata.generate.test


    

  


      

    

  

    
<img src="https://raw.githubusercontent.com/am-kantox/finitomata/master/stuff/finitomata-48x48.png" alt="Finitomata"/> Finitomata    <a href="https://kantox.com/"><img src="https://img.shields.io/badge/❤-kantox_oss-informational.svg" alt="Kantox ❤ OSS"/></a>  <img src="https://github.com/am-kantox/finitomata/workflows/Test/badge.svg" alt="Test"/>  <img src="https://github.com/am-kantox/finitomata/workflows/Dialyzer/badge.svg" alt="Dialyzer"/>
    

The FSM boilerplate based on callbacks


  
    
  
  Bird View


Finitomata provides a boilerplate for FSM implementation, allowing to concentrate on the business logic rather than on the process management and transitions/events consistency tweaking.
It reads a description of the FSM from a string in PlantUML, Mermaid, or even custom format. 

  
    
  
  Syntax Definition


Mermaid state diagram format is literally the same as PlantUML, so if you want to use it, specify syntax: :state_diagram and
if you want to use mermaid graph, specify syntax: :flowchart. The latter is the default.

Basically, it looks more or less like this

  
    
  
  PlantUML / :state_diagram


[*] --> s1 : to_s1
s1 --> s2 : to_s2
s1 --> s3 : to_s3
s2 --> [*] : ok
s3 --> [*] : ok

  
    
  
  Mermaid / :flowchart


s1 --> |to_s2| s2
s1 --> |to_s3| s3

  
    
  
  Using syntax: :flowchart


Mermaid does not allow to explicitly specify transitions (and hence event names)
from the starting state and to the end state(s), these states names are implicitly set to :*
and events to :__start__ and :__end__ respectively.

Finitomata validates the FSM is consistent, namely it has a single initial state, one or more final states, and no orphan states. If everything is OK, it generates a GenServer that could be used both alone, and with provided supervision tree. This GenServer requires to implement six callbacks
	on_transition/4 — mandatory
	on_failure/3 — optional
	on_enter/2 — optional
	on_exit/2 — optional
	on_terminate/1 — optional
	on_timer/2 — optional

All the callbacks do have a default implementation, that would perfectly handle transitions having a single to state and not requiring any additional business logic attached.
Upon start, it moves to the next to initial state and sits there awaiting for the transition request. Then it would call an on_transition/4 callback and move to the next state, or remain in the current one, according to the response.
Upon reaching a final state, it would terminate itself. The process keeps all the history of states it went through, and might have a payload in its state.

  
    
  
  Special Events


If the event name is ended with a bang (e. g. idle --> |start!| started) and
this event is the only one allowed from this state (there might be several transitions though,)
it’d be considered as determined and FSM will be transitioned into the new state instantly.
If the event name is ended with a question mark (e. g. idle --> |start?| started,)
the transition is considered as expected to fail; no on_failure/2 callback would
be called on failure and no log warning will be printed.

  
    
  
  FSM Tuning and Configuration



  
    
  
  Recurrent Callback


If timer: non_neg_integer() option is passed to use Finitomata,
then Finitomata.on_timer/2 callback will be executed recurrently.
This might be helpful if FSM needs to update its state from the outside
world on regular basis.

  
    
  
  Automatic FSM Termination


If auto_terminate: true() | state() | [state()] option is passed to use Finitomata,
the special __end__ event to transition to the end state will be called automatically
under the hood, if the current state is either listed explicitly, or if the value of
the parameter is true.

  
    
  
  Ensuring State Entry


If ensure_entry: true() | [state()] option is passed to use Finitomata, the transition
attempt will be retried with {:continue, {:transition, {event(), event_payload()}}} message
until succeeded. Neither on_failure/2 callback is called nor warning message is logged.
The payload would be updated to hold __retries__: pos_integer() key. If the payload was not a map,
it will be converted to a map %{payload: payload}.

  
    
  
  Examples


See examples directory for
real-life examples of Finitomata usage.

  
    
  
  Example


Let’s define the FSM instance
defmodule MyFSM do
  @fsm """
  s1 --> |to_s2| s2
  s1 --> |to_s3| s3
  """
  use Finitomata, fsm: @fsm, syntax: :flowchart

  ## or uncomment lines below for `:state_diagram` syntax
  # @fsm """
  # [*] --> s1 : to_s1
  # s1 --> s2 : to_s2
  # s1 --> s3 : to_s3
  # s2 --> [*] : __end__
  # s3 --> [*] : __end__
  # """
  # use Finitomata, fsm: @fsm, syntax: :state_diagram

  @impl Finitomata
  def on_transition(:s1, :to_s2, _event_payload, state_payload),
    do: {:ok, :s2, state_payload}
end
Now we can play with it a bit.
# or embed into supervision tree using `Finitomata.child_spec()`
{:ok, _pid} = Finitomata.start_link()

Finitomata.start_fsm MyFSM, "My first FSM", %{foo: :bar}
Finitomata.transition "My first FSM", {:to_s2, nil}
Finitomata.state "My first FSM"                    
#⇒ %Finitomata.State{current: :s2, history: [:s1], payload: %{foo: :bar}}

Finitomata.allowed? "My first FSM", :* # state
#⇒ true
Finitomata.responds? "My first FSM", :to_s2 # event
#⇒ false

Finitomata.transition "My first FSM", {:__end__, nil} # to final state
#⇒ [info]  [◉ ⇄] [state: %Finitomata.State{current: :s2, history: [:s1], payload: %{foo: :bar}}]

Finitomata.alive? "My first FSM"
#⇒ false
Typically, one would implement all the on_transition/4 handlers, pattern matching on the state/event.


  
    
  
  Installation


def deps do
  [
    {:finitomata, "~> 0.20"}
  ]
end

  
    
  
  Changelog


	0.27.0 — [UPD] options hibernate: boolean() and cache_state: boolean()
	0.26.0 — [UPD] a lot of tiny improvements, Finitomata.Accessible, reset_timer message + tests, experimental Finitomata.Cache
	0.25.0 — [UPD] allow assertions of entry states in Finitomata.ExUnit
	0.24.2 — [UPD/FIX] many fixes for better diagnostics in Finitomata.ExUnit 
	0.23.7 — [UPD] allow both :mox and {:mox, MyApp.Listener} as well as just MyApp.Listener as a listener in FSM definition
	0.23.4 — [FIX] many fixes to a Finitomata.ExUnit test scaffold generation
	0.23.0 — [UPD] mix finitomata.generate.test --module MyApp.FSM to generate a Finitomata.ExUnit test scaffold
	0.22.0 — [FIX] Infinitomata.start_fsm/4 is finally 102% sync
	0.21.4 — [FIX] Finitomata.Pool initialization in cluster
	0.21.3 — [FIX] proper return from Infinitomata.start_fsm/4
	0.21.1 — [UPD] listener: :mox and better Finitomata.ExUnit docs 
	0.20.2 — [UPD] allow guard matches in the RHO of ~> operator in assert_transition/3 
	0.20.0 — [FIX] starting pool on distribution, re-synch on :badrpc failure
	0.19.0 — [UPD] Finitomata.ExUnit lighten options check (compile-time module dependencies suck in >=1.16) 
	0.18.0 — [UPD] asynchronous Finitomata.Pool on top of Infinitomata 
	0.17.0 — [UPD] careful naming and Finitomata.Throttler
	0.16.0 — [UPD] Infinitomata as a self-contained distributed implementation leveraging :pg
	0.15.0 — [UPD] support snippet formatting for modern Elixir
	0.14.6 — [FIX] persistency flaw when loading [credits @peaceful-james]
	0.14.5 — [FIX] require Logger in Hook
	0.14.4 — [FIX] Docs cleanup (credits: @TwistingTwists), PlantUML proper entry
	0.14.3 — [FIX] Draw diagram in docs
	0.14.2 — [FIX] Stop Events process
	0.14.1 — [FIX] Incorrect detection of superfluous determined transitions
	0.14.0 — Finitomata.ExUnit improvements
	0.13.0 — compile-time helpers for FSM, Finitomata.ExUnit
	0.12.1 — Finitomata.on_start/1 callback
	0.11.3 — [FIX] better error message for options (credits @ray-sh)
	0.11.2 — [DEBT] exported Finitomata.fqn/2
	0.11.1 — Inspect, :flowchart/:state_diagram as default parsers, behaviour Parser
	0.11.0 — {:ok, state_payload} return from on_timer/2, :persistent_term to cache state
	0.10.0 — support for several supervision trees with ids, experimental support for persistence scaffold
	0.9.0 — [FIX] malformed callbacks had the FSM broken
	0.8.2 — last error is now kept in the state (credits to @egidijusz)
	0.8.1 — improvements to :finitomata compiler
	0.8.0 — :finitomata compiler to warn/hint about not implemented ambiguous transitions
	0.7.2 — [FIX] banged! transitions must not be determined
	0.6.3 — soft? events which do not call on_failure/2 and do not log errors
	0.6.2 — ensure_entry: option to retry a transition
	0.6.1 — code cleanup + auto_terminate: option to make :__end__ transition imminent
	0.6.0 — on_timer/2 and banged imminent transitions
	0.5.2 — state() type on generated FSMs
	0.5.1 — fixed specs [credits @egidijusz]
	0.5.0 — all callbacks but on_transition/4 are optional, accept impl_for: param to use Finitomata
	0.4.0 — allow anonymous FSM instances
	0.3.0 — en_entry/2 and on_exit/2 optional callbacks
	0.2.0 — Mermaid support

Documentation.


  

    
Finite Automata
    


  
    
  
  Above and Beyound


If you have issues fully understanding Finite Automata aka FSM, here is a one-sentence explanation that might clarify everything.
FSM consists of states, transitions, and events, where the state is an adjective (or noun,) the event is a verb and the transition is an adverbial participle.

This definition, while being mathematically lax, shallow, cursory, and maybe even perfunctory, reveals the whole and allows to grasp the very core thing about FSM nature: it describes a life in the same way the human language desribes it.

  
    
  
  Yet Another Library


I’m a big fan of Finite Automata. Always have been. Whenever we deal with a long-lived objects, they eventually have states and FSM does an enormously great job by attesting consistency, eliminating human errors and leaving the implementation with a business logic only. In a mediocre project, fifty if-then-else conditionals might perform as well as one FSM, but unless you are paid for the number of LoCs, FSM is drastically easier to carry on.

  
    
  
  Internals


This library leverages the power of callbacks to not only completely cover the FSM implementation, but also provide a compile-time proof the FSM is valid and functional. One of the most important things this library provides is the FSM description itself, that is fault-tolerant, not error-prone, and easy to grasp. The FSM definition, which is currently supported in both PlantUML and Mermaid syntaxes, would be drawn in the generated docs of the project using this library.
The consumer of this library initiates a transition by calling somewhat like transition(object, event), then the GenServer does its magic and the callback on_transition/4 gets called. From inside this callback, the consumer implements the business logic and returns the result (the next state to move the FSM to.) There are also syntactic sugar callbacks on_enter/2, on_exit/2, on_failure/3, and on_terminate/1 to allow easy state change reactions, handling of errors, and a final cleanup respectively.
All the callbacks do have a default implementation, which would perfectly handle transitions having a single to state and not requiring any additional business logic attached. When needed, this might be turned off.
Upon start, FSM moves to its initial state and sits there awaiting for the transition request. Upon this request, it’d call on_transition/4 callback and either move to the next state, or remain in the current one, according to the response from the callback. Upon reaching a final state, it would terminate itself, that’s where on_terminate/1 callback is called from. The process also keeps all the history of states it went through, and might have a payload in its state.

  
    
  
  Cool stuff


This library has a compilation-time guarantee the FSM is valid, e. g. has the only one begin state, has at least one end state, all states can be reached, and all the necessary callbacks are defined. That said, if we an FSM has an event initiating transitions from the same state to two different states, and there is no on_transition/4 clause covering that case, the compile-time error would be raised. On the other hand, if the transition is predetermined and might lead to the only one state, the callback implementation is not mandatory, because there is no trolley problem between these two states.
The FSM definition allows event names, terminated with bangs and/or question marks. If the event name is terminated with a bang (init!,) and this event is the only one possible from this state, the event will be called automatically once FSM enters this state. This is handy for moving through initialization or through states which do not require a consumer intervention and might be done immediately after FSM reaches the respective state. If the transition failed in any way (the state has not been left either due to {:error, any()} response received from on_transition/4 or due to other unexpected issue, like if on_transition/4 raised,) the on_failure/3 callback would be called and the warning would be printed to the log. To suppress this behaviour and to allow a transition silently fail, the event should have ended with a question mark  (try_call?.) The event cannot have both a bang and a question mark in its name.

  
    
  
  Wiki Example


Wikipedia provides a turnstile as an example of FSM.
[image: Turnile State Machine]
Below is the Finitomata take on this FSM.
defmodule Turnstile do
  @fsm """
  built --> |on!| locked
  locked --> |push| locked
  locked --> |coin?| unlocked
  unlocked --> |push| locked
  unlocked --> |coin?| unlocked
  unlocked --> |off| destroyed
  """

  use Finitomata, @fsm, auto_terminate: true

  def on_transition(state, :push, _event_payload, state_payload) do
    if state == :locked, do: electrocute!()
    {:ok, :locked, state_payload}
  end

  def on_transition(:locked, :coin?, _event_payload, state_payload) do
    {:ok, :unlocked, state_payload}
  end

  def on_transition(:unlocked, :coin, _event_payload, state_payload) do
    Logger.info("Thanks, this coin will be donated to the animal shelter!")
    {:error, :unexpected_coin}
  end

  def on_transition(_, :off, _, state_payload),
    do: {:ok, :destroyed, state_payload}

  # def on_failure(…), do: …

  # def on_terminate(…), do: …
end
The docs for this module would have the following diagram contained (see the ex_doc for how to enable Mermaid.)
graph TD 
  built --> |on!| locked
  locked --> |push| locked
  locked --> |coin?| unlocked
  unlocked --> |push| locked
  unlocked --> |coin?| unlocked
  unlocked --> |off| destroyed


  

    
Finitomata Compiler
    

Elixir provides an ability to introduce a custom compiler for any mix project. Finitomata leverages this feature, providing a set of diagnostics for modules declaring the FSM.

  
    
  
  Overview


Consider the following FSM.
graph TD 
  idle --> |to_s1!| s1
  s1 --> |to_s2| s2
  s1 --> |to_s3| s3
  s2 --> |to_s1| s3
  s2 --> |ambiguous| s3
  s2 --> |ambiguous| s4
  s3 --> |determined| s3
  s3 --> |determined| s4
  s4 --> |determined| s4
  s4 --> |determined| s5
Here we have transitions which do not require a handler (events to_s1!, to_s1, to_s2, and to_s3 define the target state, so just sending the event to the FSM in the respective state would be enough.) Also we have ambiguous and determined events (don’t ask about the name of the latter,) which wouldn’t succeed without the respective Finitomata.on_transition/4 handler declared. This handler must route the event to one of two possible states, conditionally returning {:ok, :s3, state} or {:ok, :s4, state} based on the state, the payload, or the event payload.
An attempt to call ambiguous event from state s2 without this handler defined would result in runtime error of the shape
{:error, {:ambiguous_transition, {:s2, :ambiguous}, [:s3. :s4]}}
but we can do better. Here is the custom compiler shining. Let’s see how the editor would have the code highlighted.
[image: Editor window with the module using Finitomata]

  
    
  
  Warnings, Infos, Hints


There are three different types of diagnostics provided by :finitomata compiler.

  
    
  
  Warning


Warning is reported both to the language server and to console during project compilation. It says “this module declares some ambiguous transitions not handled in the code.” Here is how it looks in the editor.
[image: Finitomata Warning]

  
    
  
  Info


Info is reported in cases when the compiler is not able to determine whether the ambiguous transition has been covered or not.
[image: Finitomata Warning]

  
    
  
  Hint


Hint is emitted when the compiler is positive about the ambiguous transition has been covered, just for the sake of better user experience.
[image: Finitomata Warning]

  
    
  
  WIP


This compiler is very much WIP, but it’s already fully functional, and, despite it is not able yet to report complicated cases when the transition seems to be handled, but one or more branches are unreachable, it could be a very handy tool for the additional self-check on whether the FSM is covered or the application might suffer the ambiguous, or not reachable, or not allowed runtime errors.


  

    
Finitomata behaviour
    



      

  
    
  
  Bird View


Finitomata provides a boilerplate for FSM implementation, allowing to concentrate on the business logic rather than on the process management and transitions/events consistency tweaking.
It reads a description of the FSM from a string in PlantUML, Mermaid, or even custom format. 

  
    
  
  Syntax Definition


Mermaid state diagram format is literally the same as PlantUML, so if you want to use it, specify syntax: :state_diagram and
if you want to use mermaid graph, specify syntax: :flowchart. The latter is the default.

Basically, it looks more or less like this

  
    
  
  PlantUML / :state_diagram


[*] --> s1 : to_s1
s1 --> s2 : to_s2
s1 --> s3 : to_s3
s2 --> [*] : ok
s3 --> [*] : ok

  
    
  
  Mermaid / :flowchart


s1 --> |to_s2| s2
s1 --> |to_s3| s3

  
    
  
  Using syntax: :flowchart


Mermaid does not allow to explicitly specify transitions (and hence event names)
from the starting state and to the end state(s), these states names are implicitly set to :*
and events to :__start__ and :__end__ respectively.

Finitomata validates the FSM is consistent, namely it has a single initial state, one or more final states, and no orphan states. If everything is OK, it generates a GenServer that could be used both alone, and with provided supervision tree. This GenServer requires to implement six callbacks
	on_transition/4 — mandatory
	on_failure/3 — optional
	on_enter/2 — optional
	on_exit/2 — optional
	on_terminate/1 — optional
	on_timer/2 — optional

All the callbacks do have a default implementation, that would perfectly handle transitions having a single to state and not requiring any additional business logic attached.
Upon start, it moves to the next to initial state and sits there awaiting for the transition request. Then it would call an on_transition/4 callback and move to the next state, or remain in the current one, according to the response.
Upon reaching a final state, it would terminate itself. The process keeps all the history of states it went through, and might have a payload in its state.

  
    
  
  Special Events


If the event name is ended with a bang (e. g. idle --> |start!| started) and
this event is the only one allowed from this state (there might be several transitions though,)
it’d be considered as determined and FSM will be transitioned into the new state instantly.
If the event name is ended with a question mark (e. g. idle --> |start?| started,)
the transition is considered as expected to fail; no on_failure/2 callback would
be called on failure and no log warning will be printed.

  
    
  
  FSM Tuning and Configuration



  
    
  
  Recurrent Callback


If timer: non_neg_integer() option is passed to use Finitomata,
then Finitomata.on_timer/2 callback will be executed recurrently.
This might be helpful if FSM needs to update its state from the outside
world on regular basis.

  
    
  
  Automatic FSM Termination


If auto_terminate: true() | state() | [state()] option is passed to use Finitomata,
the special __end__ event to transition to the end state will be called automatically
under the hood, if the current state is either listed explicitly, or if the value of
the parameter is true.

  
    
  
  Ensuring State Entry


If ensure_entry: true() | [state()] option is passed to use Finitomata, the transition
attempt will be retried with {:continue, {:transition, {event(), event_payload()}}} message
until succeeded. Neither on_failure/2 callback is called nor warning message is logged.
The payload would be updated to hold __retries__: pos_integer() key. If the payload was not a map,
it will be converted to a map %{payload: payload}.

  
    
  
  Examples


See examples directory for
real-life examples of Finitomata usage.

  
    
  
  Example


Let’s define the FSM instance
defmodule MyFSM do
  @fsm """
  s1 --> |to_s2| s2
  s1 --> |to_s3| s3
  """
  use Finitomata, fsm: @fsm, syntax: :flowchart

  ## or uncomment lines below for `:state_diagram` syntax
  # @fsm """
  # [*] --> s1 : to_s1
  # s1 --> s2 : to_s2
  # s1 --> s3 : to_s3
  # s2 --> [*] : __end__
  # s3 --> [*] : __end__
  # """
  # use Finitomata, fsm: @fsm, syntax: :state_diagram

  @impl Finitomata
  def on_transition(:s1, :to_s2, _event_payload, state_payload),
    do: {:ok, :s2, state_payload}
end
Now we can play with it a bit.
# or embed into supervision tree using `Finitomata.child_spec()`
{:ok, _pid} = Finitomata.start_link()

Finitomata.start_fsm MyFSM, "My first FSM", %{foo: :bar}
Finitomata.transition "My first FSM", {:to_s2, nil}
Finitomata.state "My first FSM"                    
#⇒ %Finitomata.State{current: :s2, history: [:s1], payload: %{foo: :bar}}

Finitomata.allowed? "My first FSM", :* # state
#⇒ true
Finitomata.responds? "My first FSM", :to_s2 # event
#⇒ false

Finitomata.transition "My first FSM", {:__end__, nil} # to final state
#⇒ [info]  [◉ ⇄] [state: %Finitomata.State{current: :s2, history: [:s1], payload: %{foo: :bar}}]

Finitomata.alive? "My first FSM"
#⇒ false
Typically, one would implement all the on_transition/4 handlers, pattern matching on the state/event.

  
    
  
  use Finitomata


When you use Finitomata, the Finitomata module will
do the following things for your module:
	set @behaviour Finitomata
	compile and validate FSM declaration, passed as fsm: keyword argument
	turn the module into GenServer
	inject default implementations of optional callbacks specified with
impl_for: keyword argument (default: :all)
	expose a bunch of functions to query FSM which would be visible in docs
	leaves on_transition/4 mandatory callback to be implemeneted by
the calling module and injects before_compile callback to validate
the implementation (this option required :finitomata to be included
in the list of compilers in mix.exs)



  
    
  
  Options to use Finitomata


	:fsm (String.t/0) - Required. The FSM declaration with the syntax defined by syntax option.

	:syntax - The FSM dialect parser to convert the declaration to internal FSM representation. The default value is :flowchart.

	:impl_for - The list of transitions to inject default implementation for. The default value is :all.

	:timer - The interval to call on_timer/2 recurrent event. The default value is false.

	:auto_terminate - When true, the transition to the end state is initiated automatically. The default value is false.

	:cache_state (boolean/0) - When true, the FSM state is cached in :persistent_term The default value is true.

	:hibernate - When true, the FSM process is hibernated between transitions The default value is false.

	:ensure_entry - The list of states to retry transition to until succeeded. The default value is [].

	:shutdown (pos_integer/0) - The shutdown interval for the GenServer behind the FSM. The default value is 5000.

	:persistency - The implementation of Finitomata.Persistency behaviour to backup FSM with a persistent storage. The default value is nil.

	:listener - The implementation of Finitomata.Listener behaviour or a GenServer.name() to receive notification after transitions. The default value is nil.

	:mox_envs - The list of environments to implement mox listener for The default value is [:test, :finitomata].



      


      
        Summary

    Finitomata.Accessible - Finitomata v0.27.0
    
    

    


  
  

    
Finitomata.Accessible 
    



      
The convenience module, exposing start_link/1 to embed Finitomata.Supervisor
implementation into a supervision tree in an agnostic way.
This module implements Access in a contrived way, allowing to deal with underlying
FSM instances in a following way.
iex|🌢|1 ▶ Finitomata.Accessible.start_link(type: Finitomata, implementation: L, id: L1)
{:ok, #PID<0.214.0>}

iex|🌢|2 ▶ sup = Finitomata.Accessible.lookup(L1)
%Finitomata.Accessible{
  type: Finitomata,
  id: L1,
  implementation: Finitomata.Test.Log,
  last_event: nil,
  cached_pid: #PID<0.214.0>
}

iex|🌢|3 ▶ put_in(sup, ["MyFSM"], :accept)
07:16:34.736 [debug] [→ ↹] […]
07:16:34.738 [debug] [✓ ⇄] with: [current: :*, event: :__start__, event_payload: %{payload: nil, __retries__: 1}, state: %{}]
07:16:34.738 [debug] [← ↹] […]
07:16:34.738 [debug] [→ ↹] […]
07:16:34.739 [debug] [✓ ⇄] with: [current: :idle, event: :accept, event_payload: nil, state: %{}]
07:16:34.739 [debug] [← ↹] […]

%Finitomata.Accessible{
  type: Finitomata,
  id: L1,
  implementation: Finitomata.Test.Log,
  last_event: {"MyFSM", :accept},
  cached_pid: #PID<0.214.0>
}

iex|🌢|4 ▶ get_in(sup, ["MyFSM"])
#Finitomata<[
  name: {Finitomata.L1.Registry, "MyFSM"},
  state: [current: :accepted, previous: :idle, payload: %{}],
  internals: [errored?: false, persisted?: false, timer: false]
]>

Infinitomata 
    



      
The sibling of Finitomata, but runs transparently in the cluster.
If you want to use a stateful consistent hash ring like libring,
  implement the behaviour Finitomata.ClusterInfo wrapping calls to it and 
  invoke Finitomata.ClusterInfo.init(Impl) before using Infinitomata.start_fsm/4.
The example of such an implementation for libring (assuming the named ring @ring
  has been started in the supervision tree) follows.
defmodule MyApp.ClusterInfo do
  @moduledoc false
  @behaviour Finitomata.ClusterInfo

  @impl Finitomata.ClusterInfo
  def nodes, do: HashRing.nodes(@ring)

  @impl Finitomata.ClusterInfo
  def whois(id), do: HashRing.key_to_node(@ring, id) 
end

Finitomata.ExUnit 
    



      
Helpers and assertions to make Finitomata implementation easily testable.

  
    
  
  Testing with Finitomata.ExUnit


There are several steps needed to enable extended testing with Finitomata.ExUnit.
In the first place, mox dependency should be included in your mix.exs project file
{:mox, "~> 1.0", only: [:test]}
Then, the Finitomata declaration should include a listener. If you already have the
  listener, it should be changed to Mox in :test environment, and the respecive Mox
  should be defined somewhere in test/support or like
@listener (if Mix.env() == :test, do: MyFSM.Mox, else: MyFSM.Listener)
use Finitomata, fsm: @fsm, listener: @listener
# or
# use Finitomata, fsm: @fsm, listener: {:mox, MyFSM.Listener}
If you don’t have an actual listener, the special :mox value for listener would do
  everything, including an actual Mox declaration in test environment.
use Finitomata, fsm: @fsm, listener: :mox
The last thing you need, import Mox into your test file which also does
  import Finitomata.ExUnit. That’s it, now your code is ready to use Finitomata.ExUnit
  fancy testing.

  
    
  
  Example


Consider the following simple FSM
defmodule Turnstile do
  @fsm ~S[
    ready --> |on!| closed
    opened --> |walk_in| closed
    closed --> |coin_in| opened
    closed --> |switch_off| switched_off
  ]
  use Finitomata, fsm: @fsm, auto_terminate: true

  @impl Finitomata
  def on_transition(:opened, :walk_in, _payload, state) do
    {:ok, :closed, update_in(state, [:data, :passengers], & &1 + 1)}
  end
  def on_transition(:closed, :coin_in, _payload, state) do
    {:ok, :opened, state}
  end
  def on_transition(:closed, :off, _payload, state) do
    {:ok, :switched_off, state}
  end
end
Of course, in the real life, one would not only collect the total number of passengers passed
  in the state, but also validate the coin value to let in or fail a transition, but
  for the demonstration purposes this one is already good enough.
We now want to test it works as expected. Without Finitomata.ExUnit, one would
  write the test like below
# somewhere else → Mox.defmock(Turnstile.Mox, for: Finitomata.Listener)
test "standard approach" do
  start_supervised(Finitomata.Supervisor)

  fini_name = "Turnstile_1"
  fsm_name = {:via, Registry, {Finitomata.Registry, fini_name}}

  Finitomata.start_fsm(Turnstile, fini_name, %{data: %{passengers: 0}})

  Finitomata.transition(fini_name, :coin_in)
  assert %{data: %{passengers: 0}}} = Finitomata.state(Turnstile, "Turnstile_1", :payload)

  Finitomata.transition(fini_name, :walk_in)
  assert %{data: %{passengers: 1}}} = Finitomata.state(Turnstile, "Turnstile_1", :payload)

  Finitomata.transition(fini_name, :switch_off)

  Process.sleep(200)
  refute Finitomata.alive?(Turnstile, "Turnstile_1")
end
At the first glance, there is nothing wrong with this approach, but it requires
  an enormous boilerplate, it cannot check it’s gone without using Process.sleep/1,
  but most importantly, it does not allow testing intermediate states.
If the FSM has instant transitions (named with a trailing bang, like foo!) which
  are invoked automatically by Finitomata itself, there is no way to test intermediate
  states with the approach above.
OK, let’s use Mox then (assuming Turnstile.Mox has been declared and added
  as a listener in test environment to use Finitomata)
# somewhere else → Mox.defmock(Turnstile.Mox, for: Finitomata.Listener)
test "standard approach" do
  start_supervised(Finitomata.Supervisor)

  fini_name = "Turnstile_1"
  fsm_name = {:via, Registry, {Finitomata.Registry, fini_name}}
  parent = self()

  Turnstile.Mox
  |> allow(parent, fn -> GenServer.whereis(fsm_name) end)
  |> expect(:after_transition, 4, fn id, state, payload ->
    parent |> send({:on_transition, id, state, payload}) |> then(fn _ -> :ok end)
  end)

  Finitomata.start_fsm(Turnstile, fini_name, %{data: %{passengers: 0}})

  Finitomata.transition(fini_name, :coin_in)
  assert_receive {:on_transition, ^fsm_name, :opened, %{data: %{passengers: 0}}}
  # assert %{data: %{passengers: 0}}} = Finitomata.state(Turnstile, "Turnstile_1", :payload)

  Finitomata.transition(fini_name, :walk_in)
  assert_receive {:on_transition, ^fsm_name, :closed, %{data: %{passengers: 1}}}
  # assert %{data: %{passengers: 1}}} = Finitomata.state(Turnstile, "Turnstile_1", :payload)

  Finitomata.transition(fini_name, :switch_off)
  assert_receive {:on_transition, ^fsm_name, :switched_off, %{data: %{passengers: 1}}}

  Process.sleep(200)
  refute Finitomata.alive?(Turnstile, "Turnstile_1")
end
That looks better, but there is still too much of boilerplate. Let’s see how it’d look like
  with Finitomata.ExUnit.
describe "Turnstile" do
  setup_finitomata do
    parent = self()
    initial_passengers = 42

    [
      fsm: [implementation: Turnstile, payload: %{data: %{passengers: initial_passengers}})],
      context: [passengers: initial_passengers]
    ]
  end

  test_path "respectful passenger", %{passengers: initial_passengers} do
    :coin_in ->
      assert_state :opened do
        assert_payload do
          data.passengers ~> ^initial_passengers
        end
      end

    :walk_in ->
      assert_state :closed do
        assert_payload do
          data.passengers ~> one_more when one_more == 1 + initial_passengers
        end
      end

    :switch_off ->
      assert_state :switched_off
      assert_state :*
  end
With this approach, one could test the payload in the intermediate states, and validate
  messages received from the FSM with assert_receive/3.
No other code besides assert_state/2, assert_payload/1, and ExUnit.Assertions.assert_receive/3 is
  permitted to fully isolate the FSM execution from side effects.

  
    
  
  Custom environments


In the bigger application, it might be not convenient to declare mocks for
  each and every case when Finitomata/Infinitomata might have been called under the hood.
For such cases, one might pass mox_envs: :finitomata to an FSM declaration,
  or set such a config options as config :finitomata, :mox_envs, :finitomata. That would result
  in mocks implemented for listener: :mox in this environment(s) only.
Then the tests should have been split into two groups assuming the finitomata tests
  were generated with the mix task (see below)
mix test --exclude finitomata
MIX_ENV=finitomata mix test --exclude test --include finitomata
Don’t forget to add :finitomata env to the list of envs where mox is installed

  
    
  
  Test Scaffold Generation



  
    
  
  mix tasks to simplify testing


One might generate the tests scaffold for all possible paths in the FSM with a mix task
 mix finitomata.generate.test --module MyApp.FSM

besides the mandatory --module ModuleWithUseFinitomata argument, it also accepts
--dir and --file arguments (defaulted to test/finitomata and
Macro.underscore(module) <> "_test.exs) respectively.)

Finitomata.Cache 
    



      
The self-curing cache based on Finitomata implementation.
This implementation should not be chosen for typical caching scenarios,
  use cachex and/or con_cache
  instead.
The use-case for this implementation would be somewhat like a self-updated local replica
  of the remote data. Unlike typical cache implementations, this one might keep the cached
  values up-to-date, configured by ttl: argument. Bsaed on processes (backed by Finitomata,)
  this implementation updates itself periodically, making the value retrieval almost instant.
Consider a remote service supplying currency exchange rates by polling. One might instruct
  Finitomata.Cache to retrieve values periodically (say, once per a minute,) and then
  the consumers of this cache would be able to retrieve the up-to-date values locally without
  a penalty of getting a value after a long period (cache miss.)
First of all, the Finitomata.Cache implementation should be added to a supervision tree
  {Finitomata.Cache, [
    [id: MyCache, ttl: 60_000, live?: true, type: Infinitomata, getter: &MyMod.getter/1]]}
Once the supervisor is started, the values might be retrieven as
  Finitomata.Cache.get(MyCache, :my_key_1, live?: false) # use default getter
  Finitomata.Cache.get(MyCache, :my_key, getter: fn _ -> ExtService.get(:my_key) end)

Finitomata.Pool 
    



      
The instance of FSM backed up by Finitomata.
	entry event → :__start__
	persistency → false
	listener → false
	timer → disabled
	hibernate → no
	cache_state → true


  
    
  
  FSM representation


graph TD
    * --> |__start__| idle
    idle --> |init| ready
    idle --> |do| ready
    ready --> |do| ready
    ready --> |stop| done
    done --> |__end__| *

  
    
  
  FSM paths


↝‹:* ⇥ "__start__" ↦ :idle ⇥ "init" ↦ :ready ⇥ "do" ↦ :ready ⇥ "stop" ↦ :done ⇥ "__end__" ↦ :*›
↝‹:* ⇥ "__start__" ↦ :idle ⇥ "init" ↦ :ready ⇥ "stop" ↦ :done ⇥ "__end__" ↦ :*›
↝‹:* ⇥ "__start__" ↦ :idle ⇥ "do" ↦ :ready ⇥ "do" ↦ :ready ⇥ "stop" ↦ :done ⇥ "__end__" ↦ :*›
↝‹:* ⇥ "__start__" ↦ :idle ⇥ "do" ↦ :ready ⇥ "stop" ↦ :done ⇥ "__end__" ↦ :*›

  
    
  
  FSM loops


↺‹:ready ⇥ "do" ↦ :ready›

Fully asynchronous pool to manage many similar processes, like connections.
The pool is to be started using start_pool/1 directly or with pool_spec/1 in the
  supervision tree.
initialize/2 is explicitly separated because usually this is to be done after some
  external service initialization. In a case of AMQP connection management, one
  would probably start the connection process and then a pool to manage channels.
Once initialize/2 has been called, the run/3 function might be invoked to
  asynchronously execute the function passed as actor to start_pool/1.
If the callbacks on_result/2 and/or on_error/2 are defined, they will be invoked
  respectively. Finally, the message to the calling process will be sent, unless
  the third argument in a call to run/3 is nil.

Finitomata.Throttler 
    



      
The internal definition of the call to throttle.
Finitomata.Throttler.call/3 is a blocking call similar to GenServer.call/3, but
  served by the underlying GenStage producer-consumer pair.
Despite this implementation of throttling based on GenStage is provided
  mostly for internal needs, it is generic enough to use wherever. Use the childspec
  {Finitomata.Throttler, name: name, initial: [], max_demand: 3, interval: 1_000}
  to start a throttling process and Finitomata.Throttler.call/3 to perform throttled
  synchronous calls from different processes.

  
    
  
  Usage


{:ok, pid} = Finitomata.Throttler.start_link(name: Throttler)

Finitomata.Throttler.call(Throttler, {IO, :inspect, [42]})
42

#⇒ %Finitomata.Throttler{
#   from: {#PID<0.335.0>, #Reference<0.3154300821.2643722246.59214>},
#   fun: {IO, :inspect},
#   args: ~c"*",
#   result: 42,
#   duration: 192402,
#   payload: nil
# }  

Finitomata.ClusterInfo behaviour
    



      
The behaviour to be implemented for locating the node across the cluster.
Infinitomata comes with a default naïve implementation, which simply responds
  with a determined random value for whois/1 and with all visible nodes list
  for nodes/0.
The call to Finitomata.ClusterInfo.init/1 passing the module implementing
  the desired behaviour is mandatory before any call to Infinitomata.start_fsm/4
  to preserve a determined consistency.

Finitomata.Listener behaviour
    



      
The behaviour to be implemented and passed to use Finitomata to receive
  all the state transitions notifications.

Finitomata.Parser behaviour
    



      
The behaviour, defining the parser to produce FSM out of textual representation.

Finitomata.Pool.Actor behaviour
    



      
The behaviour specifying the actor in the pool.

Finitomata.State 
    



      
Carries the state of the FSM.

Finitomata.Supervisor behaviour
    



      
THe behaviour for actual FSM implementations across Finitomata framework.
It ships with two predefined implementations, Finitomata for local
  applications and Infinitomata for distributed ones.

Finitomata.Transition 
    



      
The internal representation of Transition.
It includes from and to states, and the event, all represented as atoms.

Finitomata.Transition.Path 
    



      
The path from one state to another one

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Finitomata.Persistency - Finitomata v0.27.0
    
    

    


  
  

    
Finitomata.Persistency behaviour
    



      
The behaviour to be implemented by a persistent storage to be used
  with Finitomata (pass the implementation as persistency: Impl.Module.Name
  to use Finitomata.)
Once declared, the initial state would attempt to load the current state from
  the storage using load/1 funtcion which should return the {state, payload}
  tuple.

      


      
 